

Department of Computer Science

CSC4510 Programming Language Design and Translation

The Syntax of Pascal

Pascal Syntax

Programs and Blocks

1. program::= program heading semicolon program block

```
program hi(out); begin writeln('Hello, World!'); end.
```

2. program_heading::="program" identifier ("(" O(program_parameters) ")" |)

```
program merge(infile1, infile2, mergedfile);
```

3. program block::= block "."

```
begin writeln('Hi'); writeln('there!'); end .
```

4. block::= label_declarations constant_definitions type_definitions variable_declarations procedure function declarations statement part

```
const n=1; m=2; begin writeln(m+n); end
```

5. program parameters::= List(identifier)

```
in, out, waveItAllAbout
```

Declarations and Definitions

- 6. label declarations::= O("label" List(label) semicolon)
- 7. label::= digit sequence

```
label 1,2,3,911;
```

- 8. constant definitions::= O("const" S(constant_definition) semicolon)
- 9. constant definition::= identifier "=" constant
- 10. type definitions::= O("type" type definition semicolon #(type definition semicolon))
- 11. type definition::= identifier "=" type denoter
- 12. variable_declarations::= O("var" variable_declaration semicolon #(variable_declaration semicolon))
- 13. variable declaration::= identifier list ":" type denoter
- 14. procedure function declarations::= #(procedure declaration | function declaration) semicolon

Subprogram Declarations

15. subprogram_declaration::=procedure_declaration | function_declaration

Procedure Declarations

- 16. procedure_declaration::= procedure_heading semicolon directive | procedure_identification semicolon procedure block | procedure heading semicolon procedure block
- 17. procedure heading::= "procedure" identifier O(formal parameter list)
- 18. procedure identification::= "procedure" procedure identifier
- 19. procedure block::= block
- 20. procedure identifier::= identifier
- 21. directive::= letter #(letter | digit)
- 22. type denoter::= type identifier | new type

Function Declarations

- 23. function_declaration::= function_heading semicolon directive | function_identification semicolon function_block | function_heading semicolon function_block
- 24. function heading::= "function" identifier (formal parameter list |) ":" result type
- 25. function block::= block

Formal Parameters and Arguments

- 26. formal parameter list::= formal parameter section #(semicolon formal parameter section)
- 27. formal_parameter_section::= value_parameter_specification | variable_parameter_specification | procedural_parameter_specification | functional_parameter_specification | conformant array parameter specification
- 28. value parameter specification::= identifier list ":" type identifier
- 29. variable parameter specification::= "var" identifier list ":" type identifier
- 30. procedural parameter specification::= procedure heading
- 31. functional parameter specification::= function heading
- 32. conformant_array_parameter_specification::= value_conformant_array_specification | variable_conformant_array_specification
- 33. value_conformant_array_specification::= identifier_list ":" conformant_array_schema
- 34. variable conformant array specification::= "var" identifier list ":" conformant array schema
- 35. conformant_array_schema::= packed_conformant_array_schema | unpacked_conformant_array_schema

- 36. packed_conformant_array_schema::= "packed" "array" O(index_type_specification) "of" type_identifier
- 37. unpacked_conformant_array_schema::= "array" O(index_type_specification #(semicolon index_type_specification)) "of" type_identifier | conformant_array_schema
- 38. index type specification::= identifier ".." identifier ":" ordinal type identifier

Constants

- 39. constant::= signed number | constant identifier | character string
- 40. unsigned constant::= unsigned number | character string | constant identifier | "nil"
- 41. unsigned number::= unsigned integer | unsigned real
- 42. constant identifier::= identifier

Types

- 43. type identifier::= identifier
- 44. new type::= new ordinal type | new structured type | new pointer type
- 45. result_type::= simple_type_identifier | pointer_type_identifier
- 46. new ordinal type::= enumerated type | subrange type
- 47. new structured type::= O("packed") unpacked structured type
- 48. new pointer type::= "^" domain type
- 49. simple type identifier::= type identifier
- 50. pointer type identifier::= type identifier
- 51. enumerated type::= identifier list
- 52. subrange type::= constant ".." constant
- 53. unpacked structured type::= array type | record type | set type | file type
- 54. domain type::= type identifier
- 55. array type::= "array" #(index type #("," index type)) "of" component type
- 56. set type::= "set of" base type
- 57. file type::= "file of" component type
- 58. index type::= ordinal type
- 59. component type::= type denoter
- 60. base type::= ordinal type
- 61. ordinal type::= new ordinal type | ordinal type identifier
- 62. ordinal type identifier::= type identifier
- 63. record type::= "record" field list "end"
- 64. record section::= identifier list ":" type denoter
- 65. field list::= O(fixed part #(semicolon variant part)
- 66. fixed part::= record section #(semicolon record section)
- 67. variant part::= "case" variant selector "of" variant #(semicolon variant)
- 68. variant selector::= O(tag field ":") tag type
- 69. variant::= case constant list ":" field list
- 70. tag field::= identifier
- 71. tag_type::= ordinal_type_identifier
- 72. case constant list::= case constant #("," case constant)
- 73. case constant::= constant

Statements

```
74.
       procedure statement::= procedure identifier O(actual parameter list) | IO procedure statement
       IO procedure statement::= "read" read parameter list | "readln" readln parameter list | "write"
75.
                write parameter list | "writeln" writeln parameter list
76.
       actual parameter list::= actual parameter #( "," actual parameter)
77.
       optional file::=O(file variable ".")
78.
       read parameter list::= optional file variable access #( "," variable access)
       readln parameter list::= O( optional file variable access #( "," variable access) )
79.
80.
       write parameter list::= optional file write parameter #( "," write parameter)
81.
       writeln parameter list::= O(optional file write parameter #("," write parameter))
82.
       actual parameter::= expression | variable access | procedure identifier | function identifier
83.
       file variable::= variable access
84.
       variable access::= entire variable | component variable | identified variable | buffer varible
85.
       write parameter::= expression O( ":" O(":" expression ) )
86.
       statement part::= compound statement
87.
       compound statement::= "begin" statement sequence "end"
88.
       statement sequence::= statement #(semicolon statement)
89.
       statement::= O( label ":" ) (simple statement | structured statement)
90.
       simple statement:= empty statement | assignment statement | procedure statement |
                goto statement
91.
       structured statement::= compound statement | conditional statement | repetitive statement |
                with statement
92.
       empty statement::=
93.
       assignment statement::= variable access | function identifier ":=" expression
94.
       goto statement::= "goto" label
      conditional statement::= if statement | case statement
95.
96.
       repetitive statement::= repeat statement | while statement | for statement
97.
       loop::= repeat statement | while statement | for statement
98.
       with statement::= "with" record variable list "do" statement
99.
       if statement::= "if" boolean expression "then" statement O(else part )
100.
      case statement::= "case" case index "of" case list element #(semicolon case list element)
                O(semicolon |) "end"
101.
       repeat statement::= "repeat" statement sequence "until" boolean expression
102.
       while statement::= "while" boolean expression "do" statement
103.
      for statement::= "for" control variable ":=" initial value ("to" | "downto") final value "do"
                statement
104.
      record variable list::= record variable #("," record variable)
105.
      boolean expression::= expression
106.
      else part::= "else" statement
      case index:= expression
107.
108.
      case list element::= case constant list ":" statement
109.
      control variable::= entire variable
110.
      initial value::= expression
111.
      final value::= expression
```

Expressions and Variables

- 112. expression::= simple_expression #(relational_operator simple_expression)
- 113. function identifier::= identifier

- 114. entire variable::= variable identifier
- 115. component variable::= indexed variable | field designator
- 116. identified variable::= pointer variable "^"
- 117. buffer variable::= file variable "^"
- 118. simple expression::= O(sign) term #(adding operator term)
- 119. variable identifier::= identifier
- 120. indexed variable::= array variable "[" O(index expression #("," index expression)) "]"
- 121. field designator::= record variable "." field specifier | field designator identifier
- 122. pointer variable::= variable access
- 123. term::= factor #(multiplying operator factor)
- 124. array variable::= variable access
- 125. index expression::= expression
- 126. record variable::= variable access
- 127. field specifier::= field identifier
- 128. field designator identifier::= identifier
- 129. factor::= variable_access | unsigned_constant | function_designator | set_constructor | expression | "not" factor, | bound identifier
- 130. field identifier::= identifier
- 131. set constructor::= O(#(member designator #("," member designator)))
- 132. bound identifier::= identifier
- 133. member designator::= expression #(".." expression)
- 134. function identification::= "function" function identifier
- 135. function_designator::= function_identifier O(actual_parameter_list)

Notation

- 1. O(X) := empty | X.
- 2. #(X) :=any number of X.
- 3. S(X) := X # ("; "X).
- 4. **List**(X)::= X # (", " X).

Michael Oudshoorn August 13, 2023